
1. [bookmark: _Toc122434485][bookmark: _Toc52895930]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D22wk49 related to the title of this CR.
Contact:	Kalahasti, Narendra
	Narendra.Kalahasti@anritsu.com

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc54186268]Corrections required
2.1 Change 1
	Function name
	f_Check_EUTRA_HighFreq_PeriodicalMeasurmentReport_2Meas_8_3_1_13

	Reason for change
	Incorrect implementation of prose CR R5-225419 leads to testcase failure. Report interval of measurement report cannot be calculated after receiving just one measurement report (i.e, first MeasurementReport). Since the variable v_PreviousSFN_1 is initialized with 0, it can lead to incorrect calculation.

	Summary of change
	Report interval of measurement report is checked only after reception of second measurement report

	TTCN module
	LTE\8_3\RRC_Measurement.ttcn

	MCC160 Comment
	



Before change:
function f_Check_EUTRA_HighFreq_PeriodicalMeasurmentReport_2Meas_8_3_1_13(EUTRA_CellId_Type p_CellId,
                                                                            float p_TotalTimer,
                                                                            template (present) UL_DCCH_Message p_MeasurementReport_1,
                                                                            template (present) UL_DCCH_Message p_MeasurementReport_2,
                                                                            float p_ReportInterval_1 := 1.024,
                                                                            float p_ReportInterval_2 := 1.024) runs on EUTRA_PTC
  { //@sic R5-225419 sic@
    var SRB_COMMON_IND v_ReceivedAsp_1;
    var integer v_PreviousSFN_1 := 0;
    var integer v_CurrentSFN_1 := 0;
    var integer v_SFNLapse_1;
    
    var SRB_COMMON_IND v_ReceivedAsp_2;
    var integer v_PreviousSFN_2 := 0;
    var integer v_CurrentSFN_2 := 0;
    var integer v_SFNLapse_2;
    
    timer t_TotalTimer;
    timer t_TimeoutTimer_1;
    timer t_TimeoutTimer_2;

    t_TotalTimer.start(p_TotalTimer);
    t_TimeoutTimer_1.start(5.0);
    t_TimeoutTimer_2.start(5.0);

    alt {
      [] SRB.receive(car_SRB1_RrcPdu_IND(p_CellId, p_MeasurementReport_1)) -> value v_ReceivedAsp_1
        {
          t_TimeoutTimer_1.stop;
          t_TimeoutTimer_1.start(2.0);
          v_CurrentSFN_1 := v_ReceivedAsp_1.Common.TimingInfo.SubFrame.SFN.Number;
          
          v_SFNLapse_1 := (v_CurrentSFN_1 - v_PreviousSFN_1) mod 1024;
            
          if ((int2float(v_SFNLapse_1) > p_ReportInterval_1 * 110.0) or   // v_SFNLapse * 10 > p_ReportInterval * 110% * 1000
              (int2float(v_SFNLapse_1) < p_ReportInterval_1 * 90.0)) {    // v_SFNLapse * 10 < p_ReportInterval * 90% * 1000
            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Abnormal report interval of MeasurementReport_1!");
          }
          else {
            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Preliminary pass: periodical MeasurementReport_1 message received");
          }
          v_PreviousSFN_1 := v_CurrentSFN_1;
          repeat;
        }
      [] SRB.receive(car_SRB1_RrcPdu_IND(p_CellId, p_MeasurementReport_2)) -> value v_ReceivedAsp_2
        {
          t_TimeoutTimer_2.stop;
          t_TimeoutTimer_2.start(2.0);
          v_CurrentSFN_2 := v_ReceivedAsp_2.Common.TimingInfo.SubFrame.SFN.Number;
          
          v_SFNLapse_2 := (v_CurrentSFN_2 - v_PreviousSFN_2) mod 1024;
            
          if ((int2float(v_SFNLapse_2) > p_ReportInterval_2 * 110.0) or   // v_SFNLapse * 10 > p_ReportInterval * 110% * 1000
              (int2float(v_SFNLapse_2) < p_ReportInterval_2 * 90.0)) {    // v_SFNLapse * 10 < p_ReportInterval * 90% * 1000
            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Abnormal report interval of MeasurementReport_2!");
          }
          else {
            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Preliminary pass: periodical MeasurementReport_2 message received");
          }
          v_PreviousSFN_2 := v_CurrentSFN_2;
          repeat;
        }
      [] t_TimeoutTimer_1.timeout
        {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Timeout when receiving periodical MeasurementReport_1 message!");
        }
      [] t_TimeoutTimer_2.timeout
        {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Timeout when receiving periodical MeasurementReport_2 message!");
        }
      [] t_TotalTimer.timeout
        {
          t_TimeoutTimer_1.stop;
          t_TimeoutTimer_2.stop;
          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Periodical MeasurementReport test pass!");
        }
    }
  } //@sic R5-225419 sic@ //end of f_Check_EUTRA_HighFreq_PeriodicalMeasurmentReport_2Meas_8_3_1_13


After change:

function f_Check_EUTRA_HighFreq_PeriodicalMeasurmentReport_2Meas_8_3_1_13(EUTRA_CellId_Type p_CellId,
                                                                            float p_TotalTimer,
                                                                            template (present) UL_DCCH_Message p_MeasurementReport_1,
                                                                            template (present) UL_DCCH_Message p_MeasurementReport_2,
                                                                            float p_ReportInterval_1 := 1.024,
                                                                            float p_ReportInterval_2 := 1.024) runs on EUTRA_PTC
  { //@sic R5-225419 sic@
    var SRB_COMMON_IND v_ReceivedAsp_1;
    var integer v_PreviousSFN_1 := 0;
    var integer v_CurrentSFN_1 := 0;
    var integer v_SFNLapse_1;
    var boolean v_MeasRpt1stFlag_1 := true;
    
    var SRB_COMMON_IND v_ReceivedAsp_2;
    var integer v_PreviousSFN_2 := 0;
    var integer v_CurrentSFN_2 := 0;
    var integer v_SFNLapse_2;
    var boolean v_MeasRpt1stFlag_2 := true;
    
    timer t_TotalTimer;
    timer t_TimeoutTimer_1;
    timer t_TimeoutTimer_2;

    t_TotalTimer.start(p_TotalTimer);
    t_TimeoutTimer_1.start(5.0);
    t_TimeoutTimer_2.start(5.0);

    alt {
      [] SRB.receive(car_SRB1_RrcPdu_IND(p_CellId, p_MeasurementReport_1)) -> value v_ReceivedAsp_1
        {
          t_TimeoutTimer_1.stop;
          t_TimeoutTimer_1.start(2.0);
          v_CurrentSFN_1 := v_ReceivedAsp_1.Common.TimingInfo.SubFrame.SFN.Number;
          
          if (v_MeasRpt1stFlag_1 == true) {      // the 1st MeasurementReport message
            v_MeasRpt1stFlag_1 := false;
          }
          else {                                 // subsequent MeasurementReport messages
            v_SFNLapse_1 := (v_CurrentSFN_1 - v_PreviousSFN_1) mod 1024;
            
            if ((int2float(v_SFNLapse_1) > p_ReportInterval_1 * 110.0) or   // v_SFNLapse * 10 > p_ReportInterval * 110% * 1000
                (int2float(v_SFNLapse_1) < p_ReportInterval_1 * 90.0)) {    // v_SFNLapse * 10 < p_ReportInterval * 90% * 1000
              f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Abnormal report interval of MeasurementReport_1!");
            }
            else {
              f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Preliminary pass: periodical MeasurementReport_1 message received");
            }
          }
          v_PreviousSFN_1 := v_CurrentSFN_1;
          repeat;
        }
      [] SRB.receive(car_SRB1_RrcPdu_IND(p_CellId, p_MeasurementReport_2)) -> value v_ReceivedAsp_2
        {
          t_TimeoutTimer_2.stop;
          t_TimeoutTimer_2.start(2.0);
          v_CurrentSFN_2 := v_ReceivedAsp_2.Common.TimingInfo.SubFrame.SFN.Number;
          
          if (v_MeasRpt1stFlag_2 == true) {      // the 1st MeasurementReport message
            v_MeasRpt1stFlag_2 := false;
          }
          else {                                 //subsequent MeasurementReport messages
            v_SFNLapse_2 := (v_CurrentSFN_2 - v_PreviousSFN_2) mod 1024;
            
            if ((int2float(v_SFNLapse_2) > p_ReportInterval_2 * 110.0) or   // v_SFNLapse * 10 > p_ReportInterval * 110% * 1000
                (int2float(v_SFNLapse_2) < p_ReportInterval_2 * 90.0)) {    // v_SFNLapse * 10 < p_ReportInterval * 90% * 1000
              f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Abnormal report interval of MeasurementReport_2!");
            }
            else {
              f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Preliminary pass: periodical MeasurementReport_2 message received");
            }
          }
          v_PreviousSFN_2 := v_CurrentSFN_2;
          repeat;
        }
      [] t_TimeoutTimer_1.timeout
        {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Timeout when receiving periodical MeasurementReport_1 message!");
        }
      [] t_TimeoutTimer_2.timeout
        {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Timeout when receiving periodical MeasurementReport_2 message!");
        }
      [] t_TotalTimer.timeout
        {
          t_TimeoutTimer_1.stop;
          t_TimeoutTimer_2.stop;
          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Periodical MeasurementReport test pass!");
        }
    }
  } //@sic R5-225419 sic@ //end of f_Check_EUTRA_HighFreq_PeriodicalMeasurmentReport_2Meas_8_3_1_13

